# Generate the main makefile of the application
include("Buildings/BoundingVolumes/BVs.pro")
include("Buildings/Objects/Objects.pro")
include("Buildings/Assets/Assets.pro")

TARGET   = citybuilder
VERSION  = 1.0

# Path to the root directory of the project
DEFINES = ROOT_PATH="./"

# BVs
DIR_BV = Buildings/BoundingVolumes/
for(BV_HEADER_FILE, BV_HEADER_FILES) {
    HEADERS += $$DIR_BV$$BV_HEADER_FILE }
for(BV_SRC_FILE, BV_SRC_FILES) {
    SOURCES += $$DIR_BV$$BV_SRC_FILE }

# Objects
DIR_OBJ = Buildings/Objects/
for(OBJ_HEADER_FILE, OBJ_HEADER_FILES) {
    HEADERS += $$DIR_OBJ$$OBJ_HEADER_FILE }
for(OBJ_SRC_FILE, OBJ_SRC_FILES) {
    SOURCES += $$DIR_OBJ$$OBJ_SRC_FILE }

# Assets
DIR_ASSET = Buildings/Assets/
for(ASSET_HEADER_FILE, ASSET_HEADER_FILES) {
    HEADERS += $$DIR_ASSET$$ASSET_HEADER_FILE }
for(ASSET_SRC_FILE, ASSET_SRC_FILES) {
    SOURCES += $$DIR_ASSET$$ASSET_SRC_FILE }
for(ASSET, ASSETS) {
    INCLUDEPATH += $$DIR_ASSET/$$ASSET }

INCLUDEPATH += $$DIR_BV $$DIR_OBJ $$DIR_ASSET Architect/ ToolBox/ Map/

HEADERS += Viewer/CityViewer.hpp Architect/Architect.hpp ToolBox/Factory.hpp ToolBox/PlanParser.hpp ToolBox/LOD.hpp Map/Map.hpp
SOURCES += main.cpp Viewer/CityViewer.cpp Architect/Architect.cpp ToolBox/Factory.cpp ToolBox/PlanParser.cpp ToolBox/LOD.cpp Map/Map.cpp
OBJECTS_DIR = ./obj

CONFIG 	+= qt opengl
CONFIG  -= debug warm_on
QT	+= xml opengl

# specifications pour unix
unix {
  LIBS += -lQGLViewer -lGLEW -lGLU -lGL
}
# specifications pour windows
win32 {
  LIBS = -L"C:\Qt\libQGLViewer-2.1.1-1\QGLViewer\release" -lQGLViewer2
  CXXINCS = -I"C:\Qt\libQGLViewer-2.1.1-1\QGLViewer"
}

